4. ADA GYAKORLAT

(REKORDOK, CSOMAGOK, ÁTLÁTSZATLAN, KORLÁTOZOTT)

(matrix, verem, racionalis)

Példa - HF1.  

- Írd át a Max_Hely és a Szumma függvényeket rekurzívra!

maxhely: peldak / hf / max_kereses.adb

szumma: peldak / hf / szummazas.adb

Példa -- HF2.  

- Írj a mátrixhoz szorzás(*), skalárral való szorzás (*) és skalárral való beszorzás (procedure)

  műveleteket!

peldak /  hf /  matrixos.adb

-------------------------------------------------------------------------------------------------------------------------------------

REKORDOK

A rekord inhomogén összetett adatszerkezet, a direktszorzat és az unió típus megvalósítására szolgál.
Szintaxisa az alábbi:


TYPE <rekordnév> IS RECORD 
<adattag 1> : <típus 1> 
<adattag 2> : <típus 2> 
. 
. 
. 
<adattag n> : <típus n> 
END RECORD; 

Mezőre hivatkozás: RekordVáltNév.MezőNév

Diszkriminánsos rekord

A rekordok paraméterezhetőek, ezt hívjuk diszkriminánsnak. A diszkrimináns (paraméter) csak diszkrét típus lehet.
A szinkaxis a következő:


TYPE <rekordnév>(<diszkrimináns_név 1> : <típus 1>[:= <kezdőérték 1>][, ... ,<diszkrimináns_név n> : <típus n>[:= <kezdőérték n>]]) IS [LIMITED] RECORD 
<adattag 1> : <típus 1> 
<adattag 2> : <típus 2> 
. 
. 
. 
<adattag n> : <típus n> 
END RECORD; 

A diszkrimináns értékét a deklarációkor meg kell adni. 

Ha nincs kezdőértéke akkor kötelezően, ha van akkor opcionálisan (felül definiálható).


Diszkriminánsra hivatkozás: RekordVáltNév.DiszkrNév

Variáns rekord

 A diszkrimináns értékétől függően más-más típusú mezők szerepelhetnek benne. 

-------------------------------------------------------------------------------------------------------------------------------------

CSOMAGOK ÁLTALÁNOS HASZNÁLATA

Fordítási egység


Alprogramok  

· Eljárás:

procrdure Put_Line(text: in String);

• Függvény:

function Divider(A, B: Integer)

return Boolean;

Csomagok

• Absztrakt adattípusok

implementálása

• „Objektumok” megvalósítása

• Alprogram-könyvtárak
Csomagok

• programok tagolásának eszköze

• komponensek gyűjteménye

• összetartozó „dolgok” egységbe zárása

– Text_IO: Input-output műveletek gyűjteménye

– objektum készítéséhez

objektum-elvű programozás

– absztrakt adattípus megvalósításához

• Típusspecifikáció = típusértékhalmaz + invariáns

műveletek specifikációja

Csomag - programegység

• Specifikációs rész

külvilág számára nyújtott felület

– látható rész

kívülről használható komponensek specifikációja

– átlátszatlan rész (opcionálisan megadható)

reprezentációs információ a fordító számára

(logikailag a törzshöz tartozik)

• Törzs

megvalósítás, implementációs részletek

Specifikáció és törzs

package A is

… -- deklarációk

end A;

package body A is

… -- deklarációk

begin

… -- utasítások (pl. inicializálás)

end A;




package A is         exportált (kívülről látható)

dolgok specifikációja
procedure Eljárás( Formális: in String );

(sub)type T is …

V: Integer;

package B is … end B;

end A;

package body A is

procedure Eljárás( Formális: in String)  is … end Eljárás;

package body B is … end B;

procedure Segéd( Formális: out Integer );

(sub)type S is …            rejtett dolgok,

specifikációk és törzsek

W: Float;  

package C is … end C;

package body C is … end C;

procedure Segéd( Formális: out Integer ) is … end Segéd;

end A;

Hogyan kell használni csomag könyvtári egységet?

• a with és a use utasítás

with Text_IO;

procedure A is

begin

Text_IO.Put_Line(” Szia!”);

end;

• A használatot előre jelezni kell -with

• Mint amikor alprogramot

• A komponensekhez hozzáférek minősítéssel

• A komponensekhez való hozzáférésnél a minősítés feloldható - use

Példa1 mátrixok ---  CSOMAG
Írjunk olyan progam csomagot, amely a  megszorítás nálküli indexhatárral rendelkező mátrix típust valósítja meg a következő műveletekkel:  + és * /két mátrix közt/, * --/elemmel/, Beszoroz eljárás --- /elemmel egy mátrixot/. 
matrixok.ads

package Matrixok is

    type Index is new Integer;

    type Elem is new Integer;

    type Mátrix is array (Index range <>, Index range <>) of Elem;

    procedure Hozzáad( A: in out Mátrix; B: in Mátrix );

    function "+"( A, B: Mátrix ) return Mátrix;

    function "*" ( A, B: Mátrix ) return Mátrix;

    function "*" ( E: Elem; M: Mátrix ) return Mátrix;

    procedure Beszoroz ( E: in Elem; M: in out Mátrix );

end Matrixok;

-------------------------------------------------------------------------------------------------------------------------------------

matrixok.adb

package body Matrixok is

    procedure Hozzáad( A: in out Mátrix; B: in Mátrix ) is

    begin

        -- A := A+B;

        for I in A'Range(1) loop

            for J in A'Range(2) loop

                A(I,J) := A(I,J) + B( I-A'First(1)+B'First(1), J-A'First(2)+B'First(2) );

            end loop;

        end loop;

    end Hozzáad;

    function "+"( A, B: Mátrix ) return Mátrix is

        -- C: Mátrix := A;

        C: Mátrix(A'Range(1),A'Range(2));

    begin

        -- Hozzáad(C,B);

        -- return C;

        for I in A'Range(1) loop

            for J in A'Range(2) loop

                C(I,J) := A(I,J) + 

B( I-A'First(1)+B'First(1),

                                      J-A'First(2)+B'First(2) );

            end loop;

        end loop;

        return C;

    end "+";

    function "*" ( A, B: Mátrix ) return Mátrix is

        C: Mátrix( A'Range(1), B'Range(2) ) := (others => (others => 0));

    begin

        for I in C'Range(1) loop

            for J in C'Range(2) loop

                for K in A'Range(2) loop

                    C(I,J) := C(I,J) + 

A(I,K-B'First(1)+A'First(2)) * 

                  

B(K-A'First(2)+B'First(1),J);

                end loop;

            end loop;

        end loop;

        return C;

    end "*";

    function "*" ( E: Elem; M: Mátrix ) return Mátrix is

        C: Mátrix(M'Range(1),M'Range(2));

    begin

        for I in M'Range(1) loop

            for J in M'Range(2) loop

                C(I,J) := E * M(I,J);

            end loop;

        end loop;

        return C;

    end "*";

    procedure Beszoroz ( E: in Elem; M: in out Mátrix ) is

    begin

        for I in M'Range(1) loop

            for J in M'Range(2) loop

                M(I,J) := E * M(I,J);

            end loop;

        end loop;

    end Beszoroz;

end Matrixok;

-------------------------------------------------------------------------------------------------------------------------------------

matrixos.adb

Írjunk demo programot a matrix csomag műveleteinek használatára. Beágyazott eljárásként írjuk meg a mátrix elemeinek kiírását végző eljárást. 

with Ada.Text_IO, Matrixok;

use Ada.Text_IO, Matrixok;

procedure Matrixos is

    M: Mátrix := ((1,2),(1,2),(3,2));

    I: Mátrix := ((1,0),(0,1));

    procedure Kiír ( M: in Mátrix ) is

    begin

        for I in M'Range(1) loop

            for J in M'Range(2) loop

                Put( Elem'Image(M(I,J)) );

                Put( Ascii.HT );

            end loop;

            New_Line;

        end loop;

    end Kiír;

begin

    M := 2 * M * I;

    Beszoroz(2,M);

    Kiír(M);

end Matrixos;

-------------------------------------------------------------------------------------------------------------------------------------

Átlátszatlan típus
• A reprezentációt elrejtem a csomag specifikációjának átlátszatlan részében

• A használó programegységek nem férnek hozzá a reprezentációhoz, ezért nem is

függnek tőle

–csak a műveleteken keresztül használják a típust

• A típus részleges nézetét látják, a teljes nézet

az átlátszatlan (privát) részben van

Az átlátszatlan típus műveletei:

• A csomag által specifikált alprogramok,

– melyek valamelyik paramétere vagy visszatérési

értéke olyan típusú

• Részleges nézetből származó műveletek

   : =,  /=

Példa2  Racionálisok -- csomag, átlátszatlan 

   típus
Írjunk csomagot Pozitív racionális számok kezelésére. Magának a Pozrac típusnak a rekordban történő megvalosítását rejtsük el (rekord: számláló, nevező).

 Definiáljuk a következő függvényeket:

1.   / : egy számláló és egy nevező megadása után egy PozRac trípust ad vissza, mely egyszerűsítve van  a számláló és a nevező lnko-jával

2.  / : két Pozrac típusú szám hányadosát adja vissza, mely egyszerűsítve van  a számláló és a nevező lnko-jával

3.   / : egy PozRac és egy positive egész szám Pozrac típusú hányadosát adja vissza, mely egyszerűsítve van  a számláló és a nevező lnko-jával

4.   Számláló: egy PozRac típus számlálóját adja vissza

5.   Nevező: egy PozRac típus nevezőjét adja vissza

lnko.adb

function Lnko ( A, B : Positive ) return Positive is

   X: Positive := A;

   Y: Natural := B;

   Tmp: Natural;

begin

   while Y /= 0 loop

      Tmp := X mod Y;

      X := Y;

      Y := Tmp;

   end loop;

   return X;

end Lnko;

-------------------------------------------------------------------------------------------------------------------------------------

racionalisok.ads

package Racionalisok is

    type Racionális is private; ---átlátszatlan 
    function Számláló ( R: Racionális ) return Integer;

    function Nevező ( R: Racionális ) return Positive;

    function "/" ( Számláló: Integer; Nevező: Positive ) return Racionális;

    function "/" ( X, Y: Racionális ) return Racionális;

    function "/" ( X: Racionális; Y: Positive ) return Racionális;

    -- function "=" ( X, Y: Racionális ) return Boolean;

private

    type Racionális is record

                           Számláló: Integer := 0;

                           Nevező: Positive := 1;

                       end record;

end Racionalisok;

-------------------------------------------------------------------------------------------------------------------------------------

racionalisok.adb

with Lnko;

package body Racionalisok is

    function Számláló ( R: Racionális ) return Integer is

    begin

        return R.Számláló;

    end;

----------------------------------------------------------

    function Nevező ( R: Racionális ) return Positive is

    begin

        return R.Nevező;

    end;

----------------------------------------------------------

    function Normalizálva ( Számláló: Integer; Nevező: Positive )

    return Racionális is

    begin

        if Számláló = 0 then

            return (0,1);

        else

            declare

                H: Positive := Lnko( abs Számláló, Nevező );

            begin

                return (Számláló/H, Nevező/H);

            end;

        end if;

    end;

----------------------------------------------------------

    function "/" ( Számláló: Integer; Nevező: Positive ) return Racionális is

    begin

        return Normalizálva( Számláló, Nevező );

    end;

----------------------------------------------------------

    function "/" ( X, Y: Racionális ) return Racionális is

    begin

        return Normalizálva( X.Számláló * Y.Nevező,  X.Nevező * Y.Számláló );

    end;

----------------------------------------------------------

    function "/" ( X: Racionális; Y: Positive ) return Racionális is

    begin

        return Normalizálva( X.Számláló,  X.Nevező * Y );

    end;

----------------------------------------------------------

    -- function "=" ( X, Y: Racionális ) return Boolean is

    -- begin

    --     return X.Számláló * Y.Nevező = Y.Számláló * X.Nevező;

    -- end;

----------------------------------------------------------

end Racionalisok;

-------------------------------------------------------------------------------------------------------------------------------------

Példa3 Racionalisos.adb -- csomag, átlátszatlan







  típus
A  racionális számtípusra vonatkozó demo program  az R = 4 / 8 függvényhívást próbálja ki. Majd az R számot ossza el a  felével, és adja meg a számlálót illetve a nevezőt. 

with Racionalisok, Ada.Integer_Text_IO, Ada.Text_IO;

use Racionalisok, Ada.Integer_Text_IO, Ada.Text_IO;

procedure Racionalisos is

    R: Racionális := 4/8;

    -- X: Racionális := 3/4/5;

begin

    R := R / (R/2);

    Put( Számláló(R) );

    Put( '/' );

    Put( Nevező(R) );

end;

-------------------------------------------------------------------------------------------------------------------------------------

KORLÁTOZOTT ÁTLÁTSZATLAN TíPUS

• Nem akarjuk, hogy legyen értékadás és (nem)egyenlőségvizsgálat

– mikor egyenlő két verem?

– hogyan adunk értékül egymásnak láncolt adatszerkezettel megvalósított típusú változókat/értékeket?

• A limited kulcsszót használhatjuk az

átlátszatlan típus részleges nézetének

megadásakor

package Vermek is

subtype Elem is Integer;

type Verem (Max: Positive) is limited private;

procedure Push( V: in out Verem; X: in Elem );

…

private

type Tömb is array( Integer range <> ) of Elem;

type Verem(Max: Positive ) is record

Adatok: Tömb(1..Max);

Veremtető : Natural := 0;

end record;

end Vermek;

Példa4   Verem -- CSOMAG, KORLÁTOZOTT ÁTLÁTSZATLAN

 TíPUS
Valósítsuk meg egy diszkriminánsos rekord  (hossz a diszkrimináns, tömb az adatokank, és veremmmutató) segítségével a Verem típust és hozzá kapcsolódóan a Push, Pop,Top, Full,  Empty és Size műveleteket. Mivel két verem közt az "="  vizsgálat nehézkes, ezért  a verem adatípus legyen korlátozott (limited) átlátszatlan (private). 

vermek.ads

package Vermek is

      subtype Elem is Integer;

      type Verem (Max: Positive) is limited private;

      procedure Push( V: in out Verem; E: in Elem );

      procedure Pop( V: in out Verem; E: out Elem );

      function Top( V: Verem ) return Elem;

      function Is_Empty( V: Verem ) return Boolean;

      function Is_Full( V: Verem ) return Boolean;

      function Size( V: Verem ) return Natural;

private

      type Tömb is array( Integer range <> ) of Elem;

      type Verem(Max: Positive ) is record

                                      Adatok: Tömb(1..Max);

                                      Veremtető: Natural :=0;

                                    end record;

end Vermek;

-------------------------------------------------------------------------------------------------------------------------------------

vermek.adb

package body Vermek is

      procedure Push( V: in out Verem; E: in Elem ) is

      begin

          V.Veremtető := V.Veremtető + 1;

          V.Adatok(V.Veremtető) := E;

      end;

-------------------------------------------------------------

      procedure Pop( V: in out Verem; E: out Elem ) is

      begin

          E := Top(V);

          V.Veremtető := V.Veremtető - 1;

      end;

-------------------------------------------------------------

      function Top( V: Verem ) return Elem is

      begin

          return V.Adatok(V.Veremtető);

      end;

-------------------------------------------------------------

      function Is_Empty( V: Verem ) return Boolean is

      begin

          return V.Veremtető = 0;

      end;

-------------------------------------------------------------

      function Is_Full( V: Verem ) return Boolean is

      begin

          return V.Veremtető >= V.Max;

      end;

-------------------------------------------------------------

      function Size( V: Verem ) return Natural is

      begin

          return V.Veremtető;

      end;

end Vermek;

-------------------------------------------------------------------------------------------------------------------------------------

Példa5   VTeszt.adb -- CSOMAG, KORLÁTOZOTT 

ÁTLÁTSZATLAN TíPUS
Írjunk  demo programot verem használatára az 1 és 2 elemeket bele téve, majd a felső elemet kiiratva. 

with VERMEK, TEXT_IO; use VERMEK, TEXT_IO;

procedure VTeszt is

   V: Verem(2);

   E: Elem;

begin

        Push(v,1);

        Push(v,2);

        Pop(v,E);

        Put_LINE(ELEM'IMAGE(E));

end VTeszt;

PéldaHF

Házi feladat

- Valósítsd meg a Sor típust korlátozott átlátszatlan típusként egy

  könyvtári egységként elkészített csomagban! A reprezentációhoz használj

  diszkriminánsos rekordot és megszorítás nélküli indexhatárú tömböt!

